

A series of thin, black, overlapping lines forming various geometric shapes like triangles and polygons, located in the top-left corner of the slide.

TESTS UNITAIRES EN ELIXIR

Trucs et astuces

Le projet



Tous les matins du lundi au vendredi

Récupérer un gif aléatoire sur Les Joies du code

Et le poster dans un channel Slack

Pour commencer sa journée de travail de bonne humeur



Le cron

```
nico@otter:~$ crontab -l
```

```
SHELL=/bin/bash
```

```
0 9 * * 1,2,3,4,5 . /home/nico/.asdf/asdf.sh; cd ~/products/ljdc && mix ljdc_to_slack
```

du lundi au
vendredi à 9h

script asdf pour
mettre Elixir dans
PATH

exécution d'une tâche mix

La tâche mix

lib > mix > tasks >  ljdc._to_slack.ex > ...

```
1  defmodule Mix.Tasks.LjdcToSlack do
2    @moduledoc "Posts a random gif from LJDC in a Slack channel"
3    @shortdoc "Posts a random gif from LJDC in a Slack channel"
4
5    use Mix.Task
6
7    @impl Mix.Task
8    @spec run(any()) :: :ok
9    def run(_) do
10      Application.ensure_all_started(:ljdc)
11
12      Mix.Shell.IO.info("Fetching random gif from LJDC...")
13      {:ok, post} = LJDC.random()
14
15      Mix.Shell.IO.info("Posting gif '#{post.title}' in Slack...")
16      :ok = Slack.post_message(post)
17
18      Mix.Shell.IO.info("Enjoy !")
19    end
20  end
```

Récupération d'un gif sur les joies du code

Pas d'API -> scraping de la page web quand on clique sur le bouton "RANDOM" avec la librairie floki

```
nico@otter:~$ curl -iLs https://lesjoiesducode.fr/random
```

```
HTTP/2 302
```

```
location: https://lesjoiesducode.fr/quand-mon-code-ne-fonctionne-pas-comme-prvu
```

```
HTTP/2 200
```

```
content-type: text/html; charset=UTF-8
```

```
<!doctype html>
```

```
...
```

Une 1ère implémentation...

```
lib >  ljdk_v1.ex > ...
1  defmodule LJDC.V1 do
2    require Logger
3
4    defmodule Post do
5      | defstruct [:title, :info, :gif]
6    end
7
8    @spec random() :: {:ok, %LJDC.Post{gif: any(), info: binary(), title: any()}} | {:error, any()}
9    def random() do
10     | with {:ok, {_, 200, _}, _, html}} <- :httpc.request("https://lesjoiesducode.fr/random") do
11     |   | parse(to_string(html))
12     |   else
13     |     | error ->
14     |       | Logger.error("Failed to fetch/parse gif: #{inspect(error)}")
15     |       | {:error, error}
16     |   end
17   end
18
19 > defp parse(html) do ...
46   end
47 end
```

Une 1ère implémentation...

```
lib > ljdk_v1.ex > ...
1  defmodule LJDC.V1 do
2    require Logger
3
4  > defmodule Post do...
6    end
7
8    @spec random() :: {:ok, %LJDC.Post{gif: any(), info: binary(), title: any()}} | {:error, any()}
9  > def random() do...
17    end
18
19    defp parse(html) do
20      {:ok, document} = Floki.parse_document(html)
21
22      # get the article part
23      [article] = Floki.find(document, "article.blog-post")
24
25      # retrieve blog title
26      [{"h1", _attr, [title]}] = Floki.find(article, "h1.blog-post-title")
27
28      # retrieve blog info
29      [{"div", _attr, [_svg, span, date]}] = Floki.find(article, "div.post-meta-info")
30      {"span", _attr, [author]} = span
31
32      author =
33        case author do
34          {"a", _attr, [author]} -> author
35          author when is_binary(author) -> author
36        end
37
38      info = (author <> date) |> String.trim()
39
40      # retrieve gif url
41      [{"object", object_attr, _}] = Floki.find(article, "object")
42      {_, gif} = List.keyfind(object_attr, "data", 0)
43      result = %LJDC.Post{title: title, info: info, gif: gif}
44
45      {:ok, result}
46    end
47  end
```

Quand mon code fonctionne du premier coup et que je me rappelle que j'ai encore rien testé

🔗 Les Joies du Code, commit du 24 Mai 2024



Testons cette 1ère implementation...

```
test > ljdk_v1_test.exs > ...  
✓ 1 defmodule LJDC.V1.Test do  
2   use ExUnit.Case  
3  
✓ 4   test "nominal case" do  
5     result = LJDC.V1.random()  
6  
7     assert match?(  
8       |{:ok, %LJDC.Post{title: _, info: _, gif: _}},  
9       result  
10    )  
11  
12    {:ok, post} = result  
13    assert is_binary(post.title)  
14    assert is_binary(post.info)  
15    assert match?(%URI{}, URI.parse(post.gif))  
16  end  
17 end
```

✓ un 1er test du cas nominal
...mais c'est tout !

✗ pas sûr à 100% qu'on parse bien

✗ pas de test des cas d'erreur

✗ vraie connection HTTP à l'exécution du test
(accès internet nécessaire, le site doit être dispo,
une lenteur de la connection entraîne une lenteur du test
heureusement ce n'est pas une API HTTP payante)

✗ impossible de tester unitairement la fonction de parsing

V2 : LJDC.Parser

lib >  ljdk_parser.ex > ...

```
1 defmodule LJDC.Parser do
2   @spec parse(binary()) :: {:ok, %LJDC.Post{gif: binary(), info: binary(), title: binary()}}
3   > def parse(html) do...
30   end
31 end
```

lib >  ljdk_v2.ex > ...

```
1 defmodule LJDC.V2 do
2   require Logger
3
4   > defmodule Post do...
5   end
6
7   @spec random() :: {:ok, %LJDC.Post{gif: any(), info: binary(), title: any()}} | {:error, any()}
8   def random() do
9     with {:ok, {_ , 200, _}, _, html} <- :httpc.request("https://lesjoiesducode.fr/random") do
10      LJDC.Parser.parse(to_string(html))
11    else
12      error ->
13        Logger.error("Failed to fetch/parse gif: #{inspect(error)}")
14        {:error, error}
15    end
16  end
17 end
18 end
```

Test u de LJDC.Parser

Récupération un exemple pour les tests :

```
curl -sL "http://lesjoiesducode.fr/random" > ljdk_sample.html
```

```
test > 6 ljdk_parser_test.exs > ...
> 1 defmodule LJDC.Parser.Test do
2   use ExUnit.Case
3
4   test "nominal case" do
5     result =
6       "test/ljdk_sample.html"
7       |> File.read!()
8       |> LJDC.Parser.parse()
9
10    assert result ==
11      { :ok,
12        %LJDC.Post{
13          title: "Quand ça compile",
14          info: "Les joies du code, commit du 21 Août 2018",
15          gif: "https://lesjoiesducode.fr/content/031/v2J6obU.gif"
16        }
17      }
18  end
end
```

✓ test exhaustif et offline

Tester c'est douter.

Les Joies du Code, commit du 14 Mai 2020



Quand on me demande si j'ai bien tout testé

LSteen, commit du 3 Déc 2020



Test u de LJDC.V2

Utilisation de mocks pour tester le cas nominal et des cas d'erreur offline (bibliothèque mock)

```
test >  ljdk_v2_test.exs > {} LJDC.V2.Test >  test "nominal case"
```

```
> 1  defmodule LJDC.V2.Test do
2    use ExUnit.Case
3    import Mock
4
5    test "nominal case" do
6      html = File.read!("test/ljdc_sample.html")
7
8      with_mock :httpc, request: fn _url -> {:ok, [{"HTTP/1.1", 200, "OK"}, [], html]} end do
9        assert match?({:ok, %LJDC.Post{}}, LJDC.V2.random())
10      end
11    end
12
13    test "HTTP 500" do
14      with_mock :httpc, request: fn _url -> {:ok, [{"HTTP/1.1", 500, "Internal Server Error"}, [], ""]} end do
15        assert match?({:error, _}, LJDC.V2.random())
16      end
17    end
18
19    test "connection refused" do
20      error = {:error, {:failed_connect, [{:to_address, {"lesjoiesducode.fr", 443}}, {:inet, [:inet], :nxdomain}]}}
21
22      with_mock :httpc, request: fn _url -> error end do
23        assert match?({:error, _}, LJDC.V2.random())
24      end
25    end
26  end
```

Couplage trop fort entre le client HTTP et le test
Si on modifie le client HTTP, on doit modifier le test !


Passons à Slack et supprimons le couplage...

```
lib > slack.ex > ...
1 defmodule Slack do
2   @spec post_message(LJDC.Post.t(), atom()) :: :ok | {:error, any()}
3   def post_message(%LJDC.Post{title: title, info: info, gif: gif}, slack_client \\ Slack.Client) do
4     slack_client.post_message(%{
5       "blocks" => [
6         %{
7           "type" => "header",
8           "text" => %{
9             "type" => "plain_text",
10            "text" => title,
11            "emoji" => true
12          }
13        },
14        %{
15          "type" => "image",
16          "title" => %{
17            "type" => "plain_text",
18            "text" => info,
19            "emoji" => true
20          },
21          "image_url" => gif,
22          "alt_text" => title
23        }
24      ]
25    })
26  end
27 end
```

les-joies-du-code ▾

Quand je réalise que je me suis trompé de base pour tous mes delete

Les Joies du Code, commit du 20 Oct 2019 (437 kB) ▾



S'affranchir du couplage : contrat d'interface (behaviour) et module client


```
lib > 💧 slack_client_behaviour.ex > ...
```

```
1 defmodule Slack.Client.Behaviour do
2   @callback post_message(map()) :: :ok | {:error, any()}
3 end
```

```
lib > 💧 slack_client.ex > ...
```

```
1 defmodule Slack.Client do
2   use Tesla
3   require Logger
4
5   @behaviour Slack.Client.Behaviour
6
7   plug(Tesla.Middleware.BaseUrl, Application.get_env(:ljd, :slack_webhook))
8   plug(Tesla.Middleware.FollowRedirects)
9   plug(Tesla.Middleware.JSON)
10
11   @impl Slack.Client.Behaviour
12   def post_message(message) do
13     case post("/", message) do
14       {:ok, %Tesla.Env{status: 200}} ->
15         Logger.debug("Successfully posted message on Slack")
16         :ok
17
18       other ->
19         Logger.error("failed to post message on Slack - #{inspect(other)}")
20         {:error, "failed to post message on Slack"}
21     end
22   end
23 end
```


S'affranchir du couplage : module client de test et lib mox

```
test >  slack_v1_test.exs > ...
1  defmodule Slack.Test.V1 do
2    use ExUnit.Case
3    import Mox
4
5    setup_all do
6      # definition du module Client de test
7      Mox.defmock(Slack.Client.Mock, for: Slack.Client.Behaviour)
8
9      post = %LJDC.Post{title: "some title", info: "some info", gif: "some gif"}
10     {:ok, %{post: post}}
11   end
12
13   test "nominal case", %{post: post} do
14     # implementation du comportement (mock)
15     expect(Slack.Client.Mock, :post_message, fn message ->
16       | assert is_map(message)
17       | :ok
18     end)
19
20     assert :ok == Slack.post_message(post, Slack.Client.Mock)
21   end
22
23   test "HTTP error", %{post: post} do
24     # implementation du comportement (mock)
25     expect(Slack.Client.Mock, :post_message, fn _ ->
26       | {:error, "some error"}
27     end)
28
29     assert {:error, "some error"} == Slack.post_message(post, Slack.Client.Mock)
30   end
```

- Pas de référence à Tesla
- Nos modules clients implémentent le comportement :
pas besoin d'écrire 1 test pour chaque code d'erreur HTTP : 1 test qui retourne `{:error, xxx}` suffit !

OK mais...

Supposons qu'on change le behaviour (on retourne `:success` au lieu de `:ok`) et qu'on modifie `Slack.Client` en conséquence.

```
lib > 💧 slack_client_behaviour.ex > ...  
1  defmodule Slack.Client.Behaviour do  
2  |  @callback post_message(map()) :: :success | {:error, any()}  
3  end
```

Sans changer le test, celui-ci passe toujours !

```
• nico@otter:~/products/ljdc$ mix test test/slack_v1_test.exs  
..  
Finished in 0.02 seconds (0.00s async, 0.02s sync)  
2 tests, 0 failures  
  
Randomized with seed 904636
```



Utilisation de Hammox au lieu de Mox pour vérifier les typespec du behaviour

```
test > slack_v2_test.exs > ...  
1 defmodule Slack.Test.V2 do  
2   use ExUnit.Case  
3   import Hammox  
4  
5 > setup_all do...  
11 end  
12  
13 > test "nominal case", %{post: post} do...  
21 end  
22  
23 > test "HTTP error", %{post: post} do...  
30 end  
31 end
```

```
⊗ nico@otter:~/products/ljdc$ mix test test/slack_v2_test.exs  
Compiling 1 file (.ex)
```

```
1) test nominal case (Slack.Test.V2)  
test/slack_v2_test.exs:13  
** (Hammox.TypeMatchError)  
Returned value :ok does not match type :success | {:error, any()}.  
Value :ok does not match type :success | {:error, any()}.  
code: assert :ok == Slack.post_message(post, Slack.Client.Mock)  
stacktrace:  
  (hammox 0.7.0) lib/hammox.ex:440: Hammox.check_call/3  
  (hammox 0.7.0) lib/hammox.ex:408: Hammox.protected_code/3  
test/slack_v2_test.exs:20: (test)
```

```
.  
Finished in 0.02 seconds (0.00s async, 0.02s sync)  
2 tests, 1 failure
```

```
Randomized with seed 223518 _
```

On détecte que notre test doit être modifié :)

Bonus : tester les rollbacks ecto

Par défaut les scripts de migration ecto contiennent une callback "change/0".

Lors d'un rollback, les opérations de rollback à effectuer sont déduites de cette callback.

```
priv > repo > migrations > 20240601200435_add_posts_table.exs > ...
1  defmodule LJDC.Repo.Migrations.AddPostsTable do
2    use Ecto.Migration
3
4    def change do
5      create table(:posts) do
6        add :title, :string
7        add :info, :string
8        add :gif, :string
9        timestamps()
10     end
11   end
12 end
```

```
priv > repo > migrations > 20240601201359_delete_info_column.exs > ...
1  defmodule LJDC.Repo.Migrations.DeleteInfoColumn do
2    use Ecto.Migration
3
4    def change do
5      alter table(:posts), do: remove :info
6    end
7  end
```

Avant de lancer les scripts de migration de la base de données

🔗 Les joies du code, commit du 11 Mar 2013



Bonus : tester les rollbacks ecto

```
test > 6 ecto_rollback_test.exs > ...  
> 1  defmodule LJDC.Ecto.Rollback.Test do  
2    use ExUnit.Case  
3  
4    setup do  
5      | on_exit(fn -> Mix.Tasks.Ecto.Migrate.run([]) end)  
6    end  
7  
▷ 8    test "rollback" do  
9      | assert :ok == Mix.Tasks.Ecto.Rollback.run(["--all"])  
10     end  
11  end
```

Bonus : tester les rollbacks ecto

```
⊗ nico@otter:~/products/ljdc$ mix test test/ecto_rollback_test.exs
warning: redefining module LJDC.Repo.Migrations.DeleteInfoColumn (current version defined in memory)
priv/repo/migrations/20240601201359_delete_info_column.exs:1: LJDC.Repo.Migrations.DeleteInfoColumn (module)

warning: redefining module LJDC.Repo.Migrations.AddPostsTable (current version defined in memory)
priv/repo/migrations/20240601200435_add_posts_table.exs:1: LJDC.Repo.Migrations.AddPostsTable (module)

23:04:26.205 [info] == Running 20240601201359 LJDC.Repo.Migrations.DeleteInfoColumn.change/0 backward

23:04:26.364 [info] Migrations already up

1) test rollback (LJDC.Parser.Test)
test/ecto_rollback_test.exs:8
** (Ecto.MigrationError) cannot reverse migration command: alter table posts. You will need to explicitly define up/0 and down/0 in your migration
stacktrace:
  (ecto_sql 3.11.2) lib/ecto/migration/runner.ex:219: Ecto.Migration.Runner.execute_in_direction/5
  (elixir 1.15.6) lib/enum.ex:1693: Enum."-map/2-lists^map/1-1-"/2
  (stdlib 5.1.1) timer.erl:270: :timer.tc/2
  (ecto_sql 3.11.2) lib/ecto/migration/runner.ex:25: Ecto.Migration.Runner.run/8
  (ecto_sql 3.11.2) lib/ecto/migrator.ex:365: Ecto.Migrator.attempt/8
  (ecto_sql 3.11.2) lib/ecto/migrator.ex:325: anonymous fn/5 in Ecto.Migrator.do_down/5
  (ecto_sql 3.11.2) lib/ecto/migrator.ex:337: anonymous fn/6 in Ecto.Migrator.async_migrate_maybe_in_transaction/7
  (ecto_sql 3.11.2) lib/ecto/migrator.ex:352: Ecto.Migrator.run_maybe_in_transaction/5
  (elixir 1.15.6) lib/task/supervised.ex:101: Task.Supervised.invoke_mfa/2
  (elixir 1.15.6) lib/task/supervised.ex:36: Task.Supervised.reply/4

Finished in 0.3 seconds (0.00s async, 0.3s sync)
1 test, 1 failure

Randomized with seed 1959
```

Bonus : tester les rollbacks ecto

```
priv > repo > migrations > 20240601201359_delete_info_column.exs > {} LJDC.Repo.Migrations.DeleteInfoColumn
1  defmodule LJDC.Repo.Migrations.DeleteInfoColumn do
2    use Ecto.Migration
3
4    # def change do
5    #   alter table(:posts), do: remove :info
6    # end
7
8    def up do
9      | alter table(:posts), do: remove(:info)
10   end
11
12   def down do
13     | alter table(:posts), do: add(:info, :string)
14   end
15 end
```

Bonus : investiguer les tests random ou inter-dépendants avec `mix test --seed xxx`

test >  seed_test.exs > ...

```
1  defmodule Seed.Test do
2    use ExUnit.Case
3
4    test "test 1" do
5      | :persistent_term.put(:foo, 42)
6    end
7
8    test "test 2" do
9      | assert 42 == :persistent_term.get(:foo)
10   end
11
12   test "random" do
13     | assert Enum.random(1..5) < 2
14   end
15 end
```

- "test 2" échoue s'il est joué avant "test 1"
- "test random" échoue 4 fois sur 5

- `:seed` – an integer seed value to randomize the test suite. This seed is also mixed with the test module and name to create a new unique seed on every test, which is automatically fed into the `:rand` module. This provides randomness between tests, but predictable and reproducible results.

Bonus : investiguer les tests random ou inter-dépendants avec mix test –seed xxx

```
⊗ nico@otter:~/products/ljdc$ mix test test/seed_test.exs

1) test random (Seed.Test)
   test/seed_test.exs:12
   Assertion with < failed
   code:  assert Enum.random(1..5) < 2
   left:  5
   right: 2
   stacktrace:
     test/seed_test.exs:13: (test)

..
Finished in 0.01 seconds (0.01s async, 0.00s sync)
3 tests, 1 failure

Randomized with seed 897593
⊗ nico@otter:~/products/ljdc$ mix test test/seed_test.exs --seed 897593

1) test random (Seed.Test)
   test/seed_test.exs:12
   Assertion with < failed
   code:  assert Enum.random(1..5) < 2
   left:  5
   right: 2
   stacktrace:
     test/seed_test.exs:13: (test)

..
Finished in 0.02 seconds (0.02s async, 0.00s sync)
3 tests, 1 failure

Randomized with seed 897593 _
```

En conclusion, à retenir :

- Organiser son code pour pouvoir tester les cas nominaux et les cas d'erreurs facilement
- Mock d'une fonction interne : librairie Mock
- Mock d'un appel d'API : behaviour + librairie Hammox
- Tester les rollbacks ecto
- Savoir utiliser le seed ExUnit